Zinseszinswachstum

und

Verdopplungszeit

(*"p-mal-d" – Regel*)

Prof. Dr. J. Ziegenbalg
Institut für Mathematik und Informatik
Pädagogische Hochschule Karlsruhe

electronic mail: ziegenbalg@ph-karlsruhe.de

Literaturhinweise

I.N. Bronstein / K.A. Semendjajew / G. Musiol / H. Mühlig: Taschenbuch der Mathematik; Frankfurt a.M. 1995
R. Courant: Vorlesungen über Differential- und Integralrechnung; Berlin 1967
R. Dürr / J. Ziegenbalg: Mathematik für Computeranwendungen; Paderborn 1984

1. Einführung in die Problemstellung

■ 1.1 Zur Einstimmung: Eine Aufgabe aus der babylonischen Mathematik

Das folgende Zitat stammt aus:
Horowitz, E.: Fundamentals of Programming Languages; Springer Verlag, Berlin 1983, S. 3
siehe auch: *Knuth, D.E.*: Ancient Babylonian Algorithms; Comm. ACM, 1972

<Zitat Anfang>
Ein Schekel wird investiert. Nach wie vielen Jahren ist der Zins gleich dem investierten Kapital?
Du solltest folgendermaßen vorgehen: Berechne den Zinseszins für vier Jahre. Anfangskapital plus Zinseszins übersteigen dann 2 Schekel.
Mit welchem Faktor muß derjenige Anteil dieser Summe, der Anfangskapital plus Zinseszinsen nach 3 Jahren übersteigt, multipliziert werden, damit man den 4-Jahres-Betrag minus 2 erhält?
2 / 33 / 20 (Monate).
Ziehe 2 / 33 / 20 Monate von den vier Jahren ab, um die genaue Anzahl der Jahre und Tage zu erhalten.
Dies ist das Verfahren.
<Zitat Ende>

(Globale Rahmenbedingung: Zinssatz = 20 % pro Jahr.).

■ 1.2 Allgemeine Betrachtung in heutiger Terminologie

Ein Anfangskapital K_0 werde zum Zinssatz von p % verzinst. Die zum Jahreswechsel anfallenden Zinsen vermehren das Kapital und werden in der Folgezeit ihrerseits mitverzinst. Man spricht bei diesem Prozess vom *"Zinseszins"*-Wachstum.

Eine Grundfrage im Zusammenhang mit dem Zinseszins-Wachsum ist: Wie entwickelt sich ein gegebenes Anfangskapital bei gegebenem Zinssatz und gegebener Laufzeit?

Ein Kapital entwickelt sich bei Verzinsung im Jahresturnus entsprechend der Grundregel:

$$Kapital_{neu} = Kapital_{alt} + Zinsen$$

D.h.

$$K_1 = K_0 + K_0 \cdot \frac{p}{100} = K_0 \cdot \left(1 + \frac{p}{100}\right)$$

Insgesamt gilt somit

$$\begin{aligned}
K_1 &= \left(1 + \frac{p}{100}\right) * K_0 \\
K_2 &= \left(1 + \frac{p}{100}\right) * K_1 = \left(1 + \frac{p}{100}\right)^2 * K_0 \\
K_3 &= \left(1 + \frac{p}{100}\right) * K_2 = \left(1 + \frac{p}{100}\right)^3 * K_0 \\
&\ldots \\
K_n &= \left(1 + \frac{p}{100}\right) * K_{n-1} = \left(1 + \frac{p}{100}\right)^n * K_0
\end{aligned}$$

Das Zinseszinswachstum ist also eine spezielle Form des geometrischen Wachstums. Mit dem Wachstumsfaktor $q = 1 + \frac{p}{100}$ gilt insbesondere

$$K_n = q^n * K_0$$

---

2. Zinseszins: Ein algorithmischer Ansatz

```
Zinseszins[K0_, p_, L_] :=
  Module[{K=K0,i=0, Z},
    While[i<L,
      i=i+1; Z=K*p/100; K=K+Z;
      If[Druckoption, Print[i, "   ", Z,"   ",  K] ] ];
    Return[K] ]

Druckoption = False;
Zinseszins[100, 2.75, 20]
```

```
172.043
```

```
Druckoption=True;
Zinseszins[100, 3.5, 10]
```

```
1   3.5   103.5

2   3.6225   107.123

3   3.74929   110.872

4   3.88051   114.752

5   4.01633   118.769

6   4.1569   122.926

7   4.30239   127.228

8   4.45298   131.681

9   4.60883   136.29

10   4.77014   141.06

141.06
```

Eine Bemerkung zur Programmiermethodik: Man kann in *Mathematica* Programme im Sinne unterschiedlicher Programmierstile (imperativ, funktional, prädikativ-regelbasiert) schreiben. Das Ausdrucken von Ergebnissen ist eng mit dem imperativen Programmierparadigma verbunden. Im Sinne der Informatik ist es eine (i.a. unerwünschte, auch als Seiteneffekt bezeichnete) Nebenwirkung. Im obigen Zinseszins-Programm wurde nur deswegen davon Gebrauch gemacht, weil viele Leser dies von anderen Programmiersprachen her so kennen werden. Im folgenden soll jedoch der funktionale Programmierstil praktiziert werden.

Mathematica verfügt über eine Reihe angemessener Möglichkeiten, um Ergebnisse darzustellen; insbesondere tabellenartige und graphische Darstellungen. Davon soll im folgenden Gebrauch gemacht werden.

Tabellenartige Darstellung: Im folgenden ist nur eine allererste Demonstration gegeben; es gibt eine Fülle von Möglichkeiten, mit Hilfe von "Optionen" das Erscheinungsbild der Tabellen zu beeinflussen.

```
Druckoption = False;
Table[{L, Zinseszins[100, 2.75, L]}, {L, 0, 20}]
```

```
{{0, 100}, {1, 102.75}, {2, 105.576}, {3, 108.479}, {4, 111.462}, {5, 114.527},
 {6, 117.677}, {7, 120.913}, {8, 124.238}, {9, 127.655}, {10, 131.165},
 {11, 134.772}, {12, 138.478}, {13, 142.287}, {14, 146.199}, {15, 150.22},
 {16, 154.351}, {17, 158.596}, {18, 162.957}, {19, 167.438}, {20, 172.043}}
```

```
% //TableForm
```

0	100
1	102.75
2	105.576
3	108.479
4	111.462
5	114.527
6	117.677
7	120.913
8	124.238
9	127.655
10	131.165
11	134.772
12	138.478
13	142.287
14	146.199
15	150.22
16	154.351
17	158.596
18	162.957
19	167.438
20	172.043

Graphische Darstellung:

```
ListPlot[Table[{L, Zinseszins[100, 2.75, L]}, {L, 0, 30}],
  PlotStyle→PointSize[0.02]]
```



- Graphics -

Damit wäre ein allererster Abschluss für die Behandlung des Zinseszins-Problems erreicht.

Hat man ein Problem gelöst, so schliessen sich in der Mathematik meist in natürlicher Weise weitere Fragen an; im Falle des Zinseszins-Problems z.B. die Frage: Wie lange dauert es, bis sich das Anfangskapital verdoppelt hat? Wir landen also beim Problem der

3. Verdopplungszeit

■ Ein erster Ansatz

Aufgabe: Machen Sie sich klar, dass die Verdopplungszeit nur vom Zinssatz und nicht von der Höhe des Anfangskapitals abhängt.

```
Verdopplungszeit[p_] :=
  Module[{K=1,i=0},
    While[K < 2,
      i=i+1;  K=K*(1+p/100)];
    Return[i]  ]
```

```
Verdopplungszeit[4.5]
```

```
16
```

```
TableForm[Table[{p, Verdopplungszeit[p]}, {p, 0.5, 10, 0.5}]]
```

0.5	139
1.	70
1.5	47
2.	36
2.5	29
3.	24
3.5	21
4.	18
4.5	16
5.	15
5.5	13
6.	12
6.5	12
7.	11
7.5	10
8.	10
8.5	9
9.	9
9.5	8
10.	8

Graphische Darstellung:

```
ListPlot[Table[{p, Verdopplungszeit[p]}, {p, 0.5, 10, 0.5}],
  PlotStyle→PointSize[0.02]]
```



- Graphics -

Aufgabe: Stellen Sie anhand der Graphik Vermutungen über den (funktionalen) Zusammenhang zwischen Zinssatz und Verdopplungszeit an.

```
TableForm[
  Table[{p, Verdopplungszeit[p], p*Verdopplungszeit[p]},
    {p, 0.5, 10, 0.5}]]
```

0.5	139	69.5
1.	70	70.
1.5	47	70.5
2.	36	72.
2.5	29	72.5
3.	24	72.
3.5	21	73.5
4.	18	72.
4.5	16	72.
5.	15	75.
5.5	13	71.5
6.	12	72.
6.5	12	78.
7.	11	77.
7.5	10	75.
8.	10	80.
8.5	9	76.5
9.	9	81.
9.5	8	76.
10.	8	80.

4. Verdopplungszeit - interpoliert

■ 4.1 Veranschaulichung



Das Intervall xon etwa *x=3* bis *x=4* in vergrösserter Darstellung:




```
- Graphics -
```

■ 4.2 Verdopplungszeit - mit (linearer) Interpolation

```
VerdopplungszeitInterpoliert[p_] :=
  Module[{K=1, Kalt, Kd, i=0, d, di},
    While[K<2, i=i+1; Kalt=K; K=K*(1+p/100)];
    d=i; Kd=K;
    di=(2-Kalt)/(Kd-Kalt);
    d=(d-1)+ di;
    Return[d] ]
```

```
VerdopplungszeitInterpoliert[2.5]
```

```
28.0702
```

```
Table[{p, VerdopplungszeitInterpoliert[p]}, {p,0.5,10,0.5}]
```

```
{{0.5, 138.976}, {1., 69.6596}, {1.5, 46.5537}, {2., 35.0028}, {2.5, 28.0702},
 {3., 23.4461}, {3.5, 20.1466}, {4., 17.6687}, {4.5, 15.7431}, {5., 14.2027},
 {5.5, 12.9448}, {6., 11.8929}, {6.5, 11.0065}, {7., 10.2386}, {7.5, 9.57556},
 {8., 9.00622}, {8.5, 8.48634}, {9., 8.04147}, {9.5, 7.62709}, {10., 7.26316}}
```

```
% //TableForm
```

0.5	138.976
1.	69.6596
1.5	46.5537
2.	35.0028
2.5	28.0702
3.	23.4461
3.5	20.1466
4.	17.6687
4.5	15.7431
5.	14.2027
5.5	12.9448
6.	11.8929
6.5	11.0065
7.	10.2386
7.5	9.57556
8.	9.00622
8.5	8.48634
9.	8.04147
9.5	7.62709
10.	7.26316

```
ListPlot[Table[{p, VerdopplungszeitInterpoliert[p]},{p, 0.5, 10, 0.5}], PlotStyle→
PointSize[0.02]]
```



- Graphics -

```
TableForm[Table[{p, VerdopplungszeitInterpoliert[p],
p*VerdopplungszeitInterpoliert[p]}, {p,0.5,10,0.5}]]
```

0.5	138.976	69.4878
1.	69.6596	69.6596
1.5	46.5537	69.8305
2.	35.0028	70.0055
2.5	28.0702	70.1756
3.	23.4461	70.3383
3.5	20.1466	70.5132
4.	17.6687	70.6746
4.5	15.7431	70.8441
5.	14.2027	71.0136
5.5	12.9448	71.1963
6.	11.8929	71.3575
6.5	11.0065	71.5424
7.	10.2386	71.6699
7.5	9.57556	71.8167
8.	9.00622	72.0498
8.5	8.48634	72.1339
9.	8.04147	72.3733
9.5	7.62709	72.4574
10.	7.26316	72.6316

Die bisherigen Beobachtungen legen die folgende Vermutung nahe

Satz: Bei Zinseszinsprozessen gilt für Zinssätze zwischen $0\,\%$ und $10\,\%$ näherungsweise der folgende Zusammenhang zwischen Zinssatz p und Verdopplungszeit d: $p * d \approx 70$.

Erster Beweis (schulisch angemessen): Empirisch entsprechend der obigen Vorgehensweise ggf. mit weiteren Beispielen und Verfeinerungen.

▪ 4.3 Analyse der obigen babylonischen Aufgabe

Start:
Du solltest folgendermaßen vorgehen: Berechne den Zinseszins für vier Jahre. Anfangskapital plus Zinseszins übersteigen dann 2 Schekel.

```
Druckoption = False;
N[Table[{L, Zinseszins[1, 20, L]}, {L, 0, 4}]] // TableForm
```

0.	1.
1.	1.2
2.	1.44
3.	1.728
4.	2.0736

Kommentar:
In der Tat: Nach 4 Jahren übersteigen Anfangskapital plus Zinseszins (erstmals) 2 Schekel.

Weiter:
Mit welchem Faktor muß derjenige Anteil dieser Summe, der Anfangskapital plus Zinseszinsen nach 3 Jahren

übersteigt, multipliziert werden, damit man den 4-Jahres-Betrag minus 2 erhält?
2 / 33 / 20 Monate.





Kommentar:
"diese Summe" = `Kd`
"Anfangskapital plus Zinseszinsen nach 3 Jahren" = `Kalt`
"der Anteil dieser Summe, der Anfangskapital plus Zinseszinsen nach 3 Jahren übersteigt" = `Kd - Kalt`
"mit welchem Faktor muss ... ": `f` ... `f*(Kd - Kalt)`
"4-Jahres-Betrag minus 2" = `Kd - 2`
"mit welchem Faktor ... so dass ... ": `f*(Kd - Kalt) = Kd - 2`

Also: $f = \frac{Kd-2}{Kd-Kalt}$

Numerisch (mit dem Ergebnis in *Jahren*):

```
(2.0736 - 2) / (2.0736 - 1.728)
0.212963
```

Ergebnis in *Monaten* (im Dezimalsystem):

```
% * 12
2.55556
```

Im Dezimalsystem: 2,55556 Monate

Im 60-er System:

```
0.55556 * 60
```

33.3336

```
(0.55556 * 60 - 33) * 60
```

20.016

Also gilt näherungsweise: $2.55556 = 2 + \frac{33}{60} + \frac{20}{60^2}$

Probe in *Mathematica*:

```
2.55556 - (2 + 33 / 60 + 20 / 60^2)
```

4.44444×10^{-6}

```
% // AccountingForm
```

0.00000444444

Vgl. oben: :

```
0.016 / 60^2  // AccountingForm
```

0.00000444444

Weiter im babylonischen Text:

Ziehe 2 / 33 / 20 Monate von den vier Jahren ab, um die genaue Anzahl der Jahre und Tage zu erhalten.

```
4 - (2 + 33 / 60 + 20 / 60^2) / 12  // N
```

3.78704

Zum Vergleich:

```
VerdopplungszeitInterpoliert[20] // N
```

3.78704

5. Analyse der "*p*-mal-*d*" - Regel

■ 5.1 Der Satz von Taylor über die Reihenentwicklung

Für die Verdopplungszeit d gilt (beim Anfangskapital K_0)

$$K_d = q^d \cdot K_0 = 2 \cdot K_0.$$

Also ist $q^d = 2$

bzw. $$d = \frac{\text{Log}(2)}{\text{Log}(q)} = \frac{\text{Log}(2)}{\text{Log}(1+\frac{p}{100})} \qquad (*)$$

Kontext: Reihenentwicklung, Satz von Taylor

Satz von TAYLOR: (Zitat nach *Bronstein et. al. "Taschenbuch der Mathematik"*)
Die Funktion $y = f(x)$ sei im Intervall $[a,\ a+h]$ stetig und besitze dort stetige Ableitungen bis zur $(n-1)$-ten Ordnung. Weiterhin besitze f im Inneren des Intervalls noch die n-te Ableitung. Dann gilt

$$f(a+h) = f(a) + h\, f'(a) + \frac{h^2}{2} f''(a) + \frac{h^3}{3!} f'''(a) + \ldots + \frac{h^{n-1}}{(n-1)!} f^{(n-1)}(a) + \frac{h^n}{n!} f^{(n)}(a + \theta h)$$

mit $0 < \theta < 1$. Die Größe h kann positiv oder negativ sein.

Satz: (Zitat nach *R. Courant "Vorlesungen über Differential- und Integralrechnung"*)
Wenn die Funktion $f(x)$ in dem betrachteten Intervall stetige Ableitungen bis zur $(n+1)$-ten Ordnung besitzt, so gilt

$$f(x+h) = f(x) + h\, f'(x) + \frac{h^2}{2} f''(x) + \frac{h^3}{3!} f'''(x) + \ldots + \frac{h^n}{n!} f^{(n)}(x) + R_n$$

oder damit gleichbedeutend für $h = \xi - x$

$$f(\xi) = f(x) + (\xi - x)\, f'(x) + \frac{(\xi-x)^2}{2} f''(x) + \frac{(\xi-x)^3}{3!} f'''(x) + \ldots + \frac{(\xi-x)^n}{n!} f^{(n)}(x) + R_n\,,$$

wobei der Rest R_n durch die Formel

$$R_n = \frac{1}{n!} \int_0^h (h-\tau)^n\, f^{(n+1)}(x+\tau)\, d\tau$$

dargestellt wird.

Nach Gleichung (*) haben wir $\mathrm{Log}\left(1 + \frac{p}{100}\right)$ zu berechnen. Die Logarithmus-Funktion erfüllt die obigen Voraussetzungen für die Reihendarstellung, die in *Mathematica* durch Funktion **`Series`** realisiert wird. .

Zitat aus dem *Mathematica* Hilfe-System:
■ `Series[`f`, {`x`, `x_0`, `n`}]` generates a power series expansion for f about the point $x = x_0$ to order $(x - x_0)^n$.

Für beliebiges h erhält man:

```
Series[Log[1 + h], {h, 0, 5}]
```

$$h - \frac{h^2}{2} + \frac{h^3}{3} - \frac{h^4}{4} + \frac{h^5}{5} + O[h]^6$$

Mit $h := \frac{p}{100}$ ergibt sich daraus:

```
Series[Log[1 + h], {h, 0, 5}]   /.   h → p / 100
```

$$\frac{p}{100} - \frac{1}{2}\left(\frac{p}{100}\right)^2 + \frac{1}{3}\left(\frac{p}{100}\right)^3 - \frac{1}{4}\left(\frac{p}{100}\right)^4 + \frac{1}{5}\left(\frac{p}{100}\right)^5 + O\left[\frac{p}{100}\right]^6$$

Die Variable h entspricht im Kontext der Verdopplungszeit dem Zinssatz $\frac{p}{100}$. Der Zinssatz p liegt zwischen 0 und 10; $\frac{p}{100}$ ist also relativ klein und die höheren Potenzen dieses Terms sind vernachlässigbar.

Probe:

```
Table[Log[1 + p / 100] - p / 100, {p, 0.5, 10}]  // TableForm
```

```
-0.0000124585
-0.000111388
-0.000307387
-0.000598573
-0.000983115
-0.00145923
-0.0020252
-0.00267934
-0.00342001
-0.00424564
```

Man kann man deshalb für "kleine" Werte von p den Ausdruck $\text{Log}\left(1+\frac{p}{100}\right)$ näherungsweise durch $\frac{p}{100}$ ersetzen.

Aus der Gleichung $d = \frac{\text{Log}(2)}{\text{Log}(1+\frac{p}{100})}$ wird dann $d \approx \frac{\text{Log}(2)}{\frac{p}{100}}$. D.h. $p \cdot d \approx 100 \cdot \text{Log}[2]$.

Der numerische Wert von Log(2) ist:

```
N[Log[2]] * 100
```

```
69.3147
```

Daraus folgt: $p * d \approx \text{Log}(2) * 100 \approx 69.3 \approx 70$

Satz (*"p-mal-d"-Regel*): Bei Zinseszinsprozessen mit Zinssatz $p\,\%$ und Verdopplungszeit d gilt für Zinssätze zwischen $0\,\%$ und $10\,\%$ näherungsweise: $p * d \approx 70$.

■ 5.2 Exkurs zur Reihendarstellung in *Mathematica*

Zitat aus dem *Mathematica* Hilfe-System:

■ `Series[`f`, {`x`, `x_0`, `n`}]` generates a power series expansion for f about the point $x = x_0$ to order $(x - x_0)^n$.

```
Series[f[x],{x, 0, 5}]
```

$$f[0] + f'[0]\,x + \frac{1}{2} f''[0]\,x^2 + \frac{1}{6} f^{(3)}[0]\,x^3 + \frac{1}{24} f^{(4)}[0]\,x^4 + \frac{1}{120} f^{(5)}[0]\,x^5 + O[x]^6$$

```
Series[f[x], {x, a, 5}]
```

$$f[a] + f'[a]\,(x-a) + \frac{1}{2} f''[a]\,(x-a)^2 + \frac{1}{6} f^{(3)}[a]\,(x-a)^3 + \frac{1}{24} f^{(4)}[a]\,(x-a)^4 + \frac{1}{120} f^{(5)}[a]\,(x-a)^5 + O[x-a]^6$$

```
% /. x->a+h
```

$$f[a] + f'[a]\,((a+h)-a) + \frac{1}{2} f''[a]\,((a+h)-a)^2 + \frac{1}{6} f^{(3)}[a]\,((a+h)-a)^3 + \frac{1}{24} f^{(4)}[a]\,((a+h)-a)^4 + \frac{1}{120} f^{(5)}[a]\,((a+h)-a)^5 + O[(a+h)-a]^6$$

```
Series[Log[x],{x, 1, 5}]
```

$$(x-1)-\frac{1}{2}(x-1)^2+\frac{1}{3}(x-1)^3-\frac{1}{4}(x-1)^4+\frac{1}{5}(x-1)^5+O[x-1]^6$$

```
Series[Log[1+x],{x, 0, 5}]
```

$$x-\frac{x^2}{2}+\frac{x^3}{3}-\frac{x^4}{4}+\frac{x^5}{5}+O[x]^6$$

■ 5.3 Veranschaulichung der Reihendarstellung

```
Plot[Log[1+x], {x, -0.99, 4}, PlotRange->{{-1.1, 4},{-5,2}}]
```



- Graphics -

Visualisierung der Approximation der Logarithmus-Funktion mit Hilfe der Reihenentwicklung

```
ReihendarstellungBeispiel
```




2.5
2
1.5
1
0.5
-0.5
0.5
1
1.5
2
2.5
-0.5


2.5
2
1.5
1
0.5
-0.5
0.5
1
1.5
2
2.5
-0.5


2.5
2
1.5
1
0.5
-0.5
0.5
1
1.5
2
2.5
-0.5


2.5
2
1.5
1
0.5
-0.5
0.5
1
1.5
2
2.5
-0.5





- Graphics -

---

6. Ergänzungen

Jochen Ziegenbalg

Digital unterschrieben von Jochen Ziegenbalg
DN: cn=Jochen Ziegenbalg, c=DE, o=PH KA, ou=Inst. Math. Inf., email=ziegenbalg@ph-karlsruhe.de
Datum: 2006.06.26 18:48:31 +01'00'